Anonymous Network Channels in the Implementation of a Networked Print Spooler Application

Previously, in Listings 8, 9 and 10 we have seen that it is possible to create channels that communicate, in a transparent manner, data objects over TCP/IP networks.  We use transparent here because the processes that read and write the data objects are not aware of the whether the channel is internal to a JVM or networked.  Thus we can test systems as a set of concurrent processes running on a single processor and can then split these up into sets of processes each running on a separate processor, thereby introducing parallelism into the system.  In some situations it would be very tedious if, for example, we do not know how many networked processes are to be run to create a sufficient set of named networked channels that the system could use.  Thus the JCSP network system provides means by which anonymous networked channels can be created from distinct channel input and output ends which can be passed as parameters over the network.

To explain this capability we describe the operation of a rudimentary networked print spooling mechanism, the structure of which is shown in Figure 3 
An unknown number of PrintUser processes can be created dynamically as required.  The PrintSpooler process provides an external interface comprising two named networked channels, request and release, which are both input channels to the PrintSpooler.  A PrintUser process makes requests for service on the request channel and when the print operation has completed it will indicate this by means of a communication on the release channel.  During the spooling operation a further two anonymous networked channels are used.  The first, referred to as the useChannel, is used to send information from the PrintSpooler to the PrintUser process and the second,  printChannel, is used to send the lines to be printed to the PrintSpooler process.  The PrintSpooler has an internally set limit to the number of parallel spool operations it can process at the same time.  Thus PrintUser processes may have to wait until spooling resource is available.

Networked channels are implemented by means of the underlying socket capability.  Thus to be able to communicate, a networked writing process only needs to know the input IP address and port number allocated to a reading channel.  Implicitly this creates a networked Any2One channel.  Provided the amount of communication required to initialise a sequence of subsequent communications on another channel is small then we can set up a connection using the named channels and undertake the main communication on dynamically created anonymous channels.


[image: image1]Figure 3 The Networked Print Spooling System

The spooling system uses a single data object to pass lines to be printed from the PrintUser process to the PrintSpooler, shown in Listing 19.  The property printKey {3} is the identifier for this print job and the line {4} property is the String that forms the next line of text in the print job.

class Printline implements Serializable, JCSPCopy {

  @Property int printKey

  @Property String line

  def copy () {

    return new Printline ( printKey : this.printKey, line : this.line )

  }

}
Listing 19 The Printline Data Object

The PrintUser process is shown in Listing 20.  Its properties include the print spooler’s request and release channels {12,13} and a userId that uniquely identifies each PrintUser process {14}.  To make explanation simpler, each user will output only three lines of text as shown in the variable printList {19-22}, which is made unique by the addition of the userId to each line.  At {23} the process creates an input channel end which is assigned to the variable useChannel.  This is an anonymous input channel that is created dynamically.  It will subsequently be used by the PrintSpooler to send information to the PrintUser process.  At {24} the useChannel location is written to the printRequest channel.  This single communication indicates that this print user wants to access the print spooler and also communicates the location of the channel that can be used by the spooler to return information to the user.  At {25} the location of the print channel to be used by this print user is read.  This indicates that spooling resource is now available for the sole use of this print user.  There may be some delay between the request being made {24} and the response {25} if the spooler has no spooling resource available or has not yet been started.  The identifier, useKey,  allocated by the PrintSpooler of this spool is communicated {26}.  The PrintUser process can now make the connection to the printerChannel {28} by creating its output end.  Lines {29-30} cause the lines of the text to be sent to the spooler using the dynamically created anonymous printerChannel.  After each line has been sent the process sleeps for 10 milliseconds to ensure that any other concurrent or parallel process can interleave its output.  In reality this is not required but is included to show that the system works even when lines are being spooled from many user processes at the same time.  Once all the output has been spooled the user process sends the useKey back to the PrintSpooler on its printerRelease channel to signify that the spooling has finished and for which job.

1 class PrintUser implements CSProcess {

2   @Property ChannelOutput printerRequest

3   @Property ChannelOutput printerRelease

4   @Property int userId

5   void run() {

6     def timer = new CSTimer()

7     def printList = [ "line 1 for user ${userId}", 

8                       "line 2 for user ${userId}",

9                       "last line for user ${userId}" 

10                     ]

11     def useChannel = NetChannelEnd.createNet2One()

12     printerRequest.write(useChannel.getChannelLocation() )

13     def printChannelLocation = useChannel.read()

14     def useKey = useChannel.read()

15     println "The use key is ${useKey}"

16     def printerChannel = NetChannelEnd.createOne2Net ( printChannelLocation)    

17     printList.each { printerChannel.write (new Printline ( printKey: useKey, line: it) )

18                      timer.sleep(10) }    

19     printerRelease.write ( useKey )

20   }

21 }

Listing 20 The PrintUser process

Listing 21 shows the code for the PrintSpooler process.  Its three properties {36-38} are the printer request and release channels and the number of concurrent spoolers it will manage.  Initialisation of the variables of the process takes place {41-56}.  The variable spooling {41} holds the number of spoolers that are currently in use.  The list spoolChannels {42} will be populated with the anonymous print channels used by each spooler.  The map spoolChannelLocations {43} will hold the network location of the input end of each of these print channels using the spooler identity as the key for the map.  The list unusedSpoolers {44} holds the identifier of each spooler that is currently unused.  The map printMap {45} holds the lines of text for each spooler using the spooler identifier as the key for the map.  A simple extension to Alternative allows an array of boolean flags to be passed as a parameter of the select method.  The select method will only consider guards for which the corresponding boolean flag is true.  Thus for a guard to be chosen it must have a true flag as well as being ready to communicate or any alarm time passed, in the case of a timer guard.  The array preCon holds these flags.  We could have changed this to a list but the gain is so small that it was not considered worthwhile.  Lines {47-52} cause all these variables to be populated.  Line 47 causes an anonymous net channel input end to be created and this channel is then appended to the list of spoolChannels {48}.  Its location is then put into the spoolChannelLocations map {49}.  Note that we only create as many print channels as there are spoolers, reusing the channels for each job.  Initially, all the spoolers are available and thus each one’s identity i is appended to the list unusedSpoolers {50}.  As none of the spoolers has yet been allocated we cannot receive inputs on their associated print channels and thus the preCon element for each of the spool channels is set false {51}.  Line {53} defines a list of channels over which we will alternate subsequently.  This list contains the printerRelease and printerRequest channels and all the  spoolChannels, created at {48}.  The ALT is defined at {54} and the preCon flag for printerRelease is set true {55} as we are always willing to accept release communications.

The main loop of the run method is now entered {57} and the first operation is to set the flag, preCon[1], associated with requests for spooling that are received on printerRequest.  This is simply achieved by testing the number of spoolers in use (spooling) against the number of spoolers {58}.  Thus, if all the spoolers are in use this flag will be false and all subsequent requests for spooling will be ignored until a spooler becomes available but that depends upon the dynamics of the system.  We select the index of the channel upon which we receive an input {59} passing preCon as a parameter of the select method applied to psAlt.  The particular operation depends upon the index value {60}, over which we switch.

22 class PrintSpooler implements CSProcess {

23   @Property ChannelInput printerRequest

24   @Property ChannelInput printerRelease

25   @Property int spoolers = 2

26   void run() {

27     def spooling = 0

28     def spoolChannels = []

29     def spoolChannelLocations = [:]

30     def unusedSpoolers = []

31     def printMap = [:]

32     def preCon = new boolean[spoolers + 2 ]

33     0.upto(spoolers - 1) { i -> def c = NetChannelEnd.createNet2One()

34                                 spoolChannels << c

35                                 spoolChannelLocations[i] = c.getChannelLocation()

36                                 unusedSpoolers << i

37                                 preCon[i+2] = false

38                          }

39     def altChans = [ printerRelease, printerRequest, * spoolChannels ]

40     def psAlt = new ALT ( altChans )
41     preCon[0] = true

42     while (true) {

43       preCon[1] = (spooling < spoolers)

44       def index = psAlt.select(preCon)

45       switch (index) { 

46         case 0:

47           def usedKey = printerRelease.read()

48           unusedSpoolers.add(usedKey)

49           preCon[usedKey + 2] = false

50           spooling = spooling - 1

51           def lines = printMap.get(usedKey)

52           println "\n\nUse key was ${usedKey}\n\n"

53           lines.each{ println "${it}" }

54           println "================================"

55           printMap.remove(usedKey)

56           break

57         case 1:

58           def useChannelLocation = printerRequest.read()

59           def useChannel = NetChannelEnd.createOne2Net(useChannelLocation)

60           spooling = spooling + 1

61           def useKey = unusedSpoolers.pop()

62           preCon[useKey+2] = true

63           printMap[useKey] = [] // initialise the printlist for this user

64           useChannel.write(spoolChannelLocations.get(useKey) )

65           useChannel.write( useKey )

66           break

67         default :

68           def pLine = spoolChannels[ index - 2].read()

69           printMap[pLine.printKey] << pLine.line

70       } //switch

71     } //while

72   } // run

73 } // class

Listing 21 The PrintSpooler process

Case 0 {61-71} deals with a user releasing their use of a spooler.  From the printerRelease channel we can read {62} the usedKey that was allocated to this spool job.  For this explanatory example we have simply used the identity of the spooler.  This spooler can now be added into the list of unusedSpoolers {63}.  The preCon flag associated with this spooler can now be set false as we can no longer receive valid lines from this user {64}.  The number of spooling spoolers can be decremented {65}.  The lines of text associated with this usedKey can then be extracted from the printMap {66}, an appropriate header for the text printed {67} and then each of the lines printed {68}, followed by a banner {69}.  Finally, we remove this entry from the printMap {70}.

Case 1 {72-81} deals with requests for printing, recall that such requests will only be read by this process when it has a spooler available and this is tested at the start of the main loop {58}.  The location of the useChannel, by which the PrintSpooler outputs information to the requesting PrintUser is read {73} and then an anonymous network output channel end is created using this location {74}.  The number of spooling spoolers in use is incremented {75} and we pop the identity (useKey) of an unused spooler form the list of unusedSpoolers {76}.  The preCon flag associated with the spooler is then set true {77} and the list associated with the useKey in the printMap entry is set empty {78}.  Finally, the print channel location and spool job identity are sent to the PrintUser by writing the values to the useChannel {79-80}.  The spooler is now ready to receive lines of text to put in the map and the PrintUser has all the information it requires to both initialise the print channel it is to use and has an identity for the job it is about to start.

The default case {82-84} simply contains the reading of lines of text into allocated spoolers from PrintUsers, recalling that the preCon flags ensure that we only read lines from PrintUsers that have been allocated a printChannel and an entry in the printMap.  A line is read from the printChannel {83} and placed in the printMap {84}, recall that the job identity is the same as the spooler identity, which means that many jobs will have the same job number.  It would be a simple extension for the interested reader to create a further map that allocated unique job numbers to the spoolers.

Listing 22 shows an instance of the PrintUser process invocation.  Two named network channels REQUEST and RELEASE are defined as Any2Net implying that we can have many such PrintUser process invocations.  The userId of the PrintUser is obtained from a command line parameter.

74 Node.getInstance().init(new TCPIPNodeFactory ())

75 def pRequest = CNS.createAny2Net ("REQUEST")

76 def pRelease = CNS.createAny2Net ("RELEASE")

77 new PAR ( [ new PrintUser ( printerRequest: pRequest, 

78                             printerRelease: pRelease, 

79                             userId : args[0].toInteger() 

80                           )

81           ] ).run()

Listing 22 The Script to execute a PrintUser Process

Listing 23 gives the required invocation of the single PrintSpooler process, which yet again defines the net channels REQUEST and RELEASE as input channels from the network to this process.  The number of spoolers managed by the process is specified by means of a command line argument.

82 Node.getInstance().init(new TCPIPNodeFactory ())

83 def pRequest = CNS.createNet2One ("REQUEST")

84 def pRelease = CNS.createNet2One ("RELEASE")

85 new PAR ( [ new PrintSpooler ( printerRequest: pRequest, 

86                                printerRelease: pRelease, 

87                                spoolers : args[0].toInteger()  

88                              )

89           ] ).run()

Listing 23 The Script to execute the PrintSpooler Process

Output from the spooler process is shown in Output 3 and results from the invocation of three PrintUser processes and one PrintSpooler managing two spoolers.  All the user processes were started, in numerical sequence and waiting before the spooler process was invoked.  It can be seen that spooler zero was used twice and that the order in which the users were serviced was not related to the order in which they were started.  Subsequently another PrintUser process was invoked (not shown) and its output appeared in the PrintSpooler window as expected, implying that the spooling service could exist on a network being used as required.

Use key was 0

line 1 for user 2

line 2 for user 2

last line for user 2

================================

Use key was 1

line 1 for user 3

line 2 for user 3

last line for user 3

================================

Use key was 0

line 1 for user 1

line 2 for user 1

last line for user 1

================================
Output 3 Output from the PrintSpooler process

